Paper 182-31

# Empirical Power Estimation for Phase I Dose Proportionality Studies Based on Power-Law Model Using Confidence Interval Criteria

Jihao Zhou[§], Jane Li, Bruce Coate
Pfizer Global Research and Development La Jolla Laboratories, San Diego, CA
Melissa Chen
Allergan, Inc., Irvine, CA

## ABSTRACT

A Phase I dose-proportionality study is an essential tool to understand drug pharmacokinetic dose-response relationship in early clinical development. There are a number of different approaches to the assessment of dose proportionality. The confidence interval (CI) criteria approach, a staitistically sound and clinically relevant approach, has been proposed to detect dose-proportionality (Smith, et al. 2000), by which the proportionality is declared if the 90% CI for slope is completely contained within the pre-determined critical interval. This method, enhancing the information from a clinical dose-proportionality study, has gradually drawn attention. However, exact power calculation of dose proportinality studies based on CI criteria poses difficulty for practioners since the methodology was essentailly from two one-sided tests (TOST) procedure for the slope, which should be unit under proportionality. It requires sophisticated numerical integration, and it is not available in statistical software packages. This paper presents a SAS Macro to compute the empirical power for the CI-based dose proportinality studies. The resulting sample sizes and corresponding empirical powers suggest that this approach is powerful in detecting dose-proportionality under commonly used sample sizes for phase I studies.

**Keywords**: dose-proportinality, empirical power, power-law model, pharmacokinetics, mixed-effects model

## INTRODUCTION (HEADER 1)

A Phase I dose-proportionality study is an essential tool to understand drug dose-response relationship in early clinical development. The relationship between the administered dose of a drug and its therapeutic effects is, therefore, of major interest in drug development. Proportionality indicates that doubling dose doubles the drug exposure, measured by the concentration-related phamacokinetic (PK) parameters such as the maximum plasma concentration ($C_{max}$) and the area under the plasma concentration time curve (AUC). Mathematically, proportinality requires an intercept of zero as well as a linear relationship between dose and $PK = c \cdot Dose$. Numerous statistical analyses are available for assessing dose-proportionality. One simple approach is to do regression of untransformed PK on dose:

$$PK = b_0 + b_1 \cdot Dose + b_2 \cdot Dose^2 + e \quad (1)$$

and then test for statistically nonsignificant intercept and quardratic terms. However, the power of this method and the posibility that the study was incapable of detecting important departures from proportinality make this approach rarely used. A typical dose-proportionality study consists of a crossover design with D doses being given to N subjects over P periods. Commonly D and P are same. Analysis of variance is another approach:

$$\log(PK / D_k) = s_i + p_j + q_k + e_{ijk} \quad i = 1,..N;\ j = 1,..P;\ k = 1,..D \quad (2)$$

It would be assumed that PK was log-normally distributed with equal variances (on the log scale) at different dose levels. The overall F-test for equality of mean values across doses in this dose-divide model is used for assessing the proportionality. This test suffers from the same drawbacks as the quardratic regression approach. Another one, an empirical model relating the log of the PK parameter linearly to the log of the dose, is called “power-law model” :

$$PK = c \cdot Dose^{b_1} \cdot e^{e} \quad (3)$$

Dose-proportionality implies that $b_1$ =1. Compared to the analysis of variance approach, it has a gain in power because of the simultaneous use of data from all doses. Literature has shown that power-law model seems to adequatedly describe the relationship between PK parameters and dose over the dose range examined even in studies with relatively few doses(three or four) [1]. A CI approach has been proposed to assess dose

[§]Current affiliation: Department of Biostatistics, Allergan, Inc., 2525 Dupont Drive, CA 92612.

proportionality based on the power-law model [2], and now gradually become popular.

The exact power calculation for the dose-proportionality study based on the CI-based approach requires sophisticated numerical integration and it is not available in statistical software packages. A Monte Carlo simulation is a useful alternative approach to the power calculation in this particular situation [3]. This paper presents SAS Macros to compute the empirical power of the cross-over designs under different scenarios in dose-proportionality studies using Monte Carlo simulations.

## BACKGROUND

### CONFIDENCE INTERVAL CRITERIA APPROACH FOR ASSESSING DOSE-PROPORTIONALITY

After the logarithmic transformation, the power-law model can be expressed as:

$$\log(PK) = b_0 + b_1 \cdot \log(Dose) + e \quad (4)$$

The predicted geometric mean of the highest dose is $e^{b_0} h^{b_1}$ and that of the lowest dose is $e^{b_0} l^{b_1}$. Dose-proportionality corresponds to $e^{b_0} h^{b_1} / e^{b_0} l^{b_1} = h/l$, which can be rewritten as: $(h/l)^{b_1 - 1} = r^{b_1 - 1} = 1$. As an analogy to the bioequivalence test, lower and upper limits are defined as: $q_L < r^{b_1 - 1} < q_H$, equivalently, $\log(q_L) < (b_1 - 1)\log(r) < \log(q_H)$. Then the critical interval is given as follows:

$$1 + \frac{\log(q_L)}{\log(r)} < b_1 < 1 + \frac{\log(q_H)}{\log(r)} \quad (5)$$

The commonly used ($q_L$, $q_U$) are (80%, 125%) and (77%, 130%). The type-I error $\alpha$ of the procedure is often set as 5%. Dose proportionality would be declared when the $(1-2\alpha)\cdot 100\%$ CI for $b_1$ lies entirely within the critical interval.

Therefore, demonstrating proportionality is essentially an equivalence problem:

$H_{01}$: $b_1 \le 1 + \frac{\log(q_L)}{\log(r)}$ or $H_{02}$: $b_1 \ge 1 + \frac{\log(q_H)}{\log(r)}$

$H_a$: $1 + \frac{\log(q_L)}{\log(r)} < b_1 < 1 + \frac{\log(q_H)}{\log(r)}$

### MIXED-EFFECTS MODEL

When repeated measures present in a dose-proportionality study, a mixed-effects statistical model based on Eq. (4) can be used to account for correlation between correlated measurements in a given subject. The following general form was considered:

$$\log(PK_{ij}) = (b_0 + h_{1i}) + (b_1 + h_{2i}) \cdot \log(Dose_{ij}) + e_{ij} \quad (6)$$

for the $j^{th}$ observation of PK in the $i^{th}$ subject. The model denotes random between-subject variability $h_{1i} \sim N(0, x_1^2)$ in the intercept parameter, random between-subject variability $h_{2i} \sim N(0, x_2^2)$ in the slope, as well as random $e_{ii} \sim N(0, s^2)$. The data may not support inclusion of $h_{2i}$ in the model, in other words, it may be appropriate to consider a constant slope. The random effects, as well as the fixed effects ($b_0$ and $b_1$) and their 90% CIs, can be estimated with the MIXED procedure of SAS.

### CROSSOVER DESIGNS

Crossover designs have been widely used in drug clinical trials [4,5]. In a crossover trial, subjects are given sequences of treatments with the object of studying differences between individual treatments [6]. A Latin square, each dose (treatment) being represented once and only once in each period and each sequence, is a special case in crossover designs. The simulation results for a 3x3 Latin square design will be presented.

## METHODS

### STATISTICAL SIMULATIONS

The Monte Carlo simulation involves random sampling techniques to generate a series of random samples from a distribution that represents the study population of interest (eg, the population under $H_a$, or dose-proportionality exists). For each generated random sample, the CI-based dose-proportionality assessment procedure is applied and the conclusion of rejecting or accepting $H_0$ is made. The empirical power of the assessment procedure is then calculated as the proportion of the replications in which $H_0$ is rejected. Herein, we generate *m* (eg, 1000) samples of size *n* (eg, 12). To save the computation time, we generate all these *m* samples by calling a multivariate normal random number generator (a SAS Macro) once, then analyze the samples in sequential orders. The computational process is outlined as follows:

Step 1. Generate *m* random samples of size *n* according to the following multivariate nomal distribution with pre-specified mean vector $m$ and covariance matrix $\Sigma$,

$$X \sim MVN(m, \Sigma)$$

Step 2. Run PROC MIXED on the *i*th random sample to calculate the 100 (1-2α) % CI for $b_1$, for *i*=1 to *m*.

Step 3. The empirical power is calculated as the proportion of *m* random samples in which the CI falls entirely within pre-specified critical interval: $(1+\log(q_L)/\log(r),\ 1+\log(q_H)/\log(r))$.

**SAS PROGRAM DEVELOPMENT**

**I). 3x3 Latin square ( 3-sequence, 3-Period Crossover Design )**

First, the macro deletes data set *empower,* which is used to store coverage results, if any exists from previous macro calls. The SAS Macro MVN() [7] is called to generate *(mn)* random normal trivariates. *(mn)/3* random normal trivariates for the first sequence (ABC), *(mn)/3* random normal trivariates for the second sequence (BCA) and *(mn)/3* random normal trivariates for the third sequence (CAB) are based on the multivariate normal distribution with mean vector $m$ and compound symmetric covariance matrix $\Sigma$ under $H_a$. A, B, and C denote 3 dose levels (eg, A=200, B=400, C=700). The mean vector $m$ is determined by the sequence of doses, for example, the samples in the first sequence have $m$ = (log(A), log(B), log(C)). The covariance matrix $\Sigma$ is specified by the proposed vaules for correlation coefficient between repeated measures (ρ) and standard deviation for the measurements (σ). These random trivariates are stored in a dataset called *samples*. For the $i^{th}$ iteration of total *m*, a SAS dataset step selects the $i^{th}$ sample from the dataset *samples*, then PROC TRANSPOSE and a SAS dataset step are used to rearrange the data in a format that is suitable for PROC MIXED. The 100 (1-2α) % CI for the direct treatment effect is obtained by applying PROC MIXED on the $i^{th}$ sample and saved into a dataset *empower*. PROC APPEND is then used to concatenate all *m* resulting data sets. SAS data step is used to produce a binary (0/1) variable to indicate whether the CI from each simulation falls within the criteria $(1+\log(q_L)/\log(r),\ 1+\log(q_H)/\log(r))$. If it does, it means the $H_0$ is rejected. Finally, SAS PROC MEANS is used to obtain the proportion of rejecting $H_0$ for all simulations (ie, the empirical power of the assessment procedure), and keep the result in dataset *power*.

```
%Macro DOSE_PROP3x3 (runs=1000, beta=1.0, Means=200 400 700, SD=0.338, N=12, r=0.40,
alpha=0.10, lower=0.80, upper=1.25, rr=700/200);

* prepare the SAS dataset;
      proc datasets library=work nolist;
      delete empower;
      run;
      quit;

 * Store the variance-covariance matrix in a data set ;
 data varcov;
       m1=&sd*&sd;     m2=&sd*&sd*&r;  m3=&sd*&sd*&r;   output;
      m1=&sd*&sd*&r;  m2=&sd*&sd;     m3=&sd*&sd*&r;   output;
      m1=&sd*&sd*&r;  m2=&sd*&sd*&r;  m3=&sd*&sd;      output;
      ;

* Store the mean vector in a data set ;
%let A=%scan(&means,1, " ");
%let B=%scan(&means,2, " ");
%let C=%scan(&means,3, " ");

data means;
      m1=log(1.0*symget('A')) *&beta; output;
      m1=log(1.0*symget('B')) *&beta; output;
      m1=log(1.0*symget('C')) *&beta; output;
      ;
```

```
run;

**** Monte Carlo simulation starts ***;
* Simulate multivariate normal variables;
* For all iterations by using SAS Macro MVN (SAS Technical Support);

%mvn(varcov=varcov,means=means,n=%eval(&n*&runs),sample=b, seed=0)

* Add an iteration indicator variable (Iteration) into the generated data set;
data b;
set b;
Iteration=int((_n_-1)/&n);
run;

%do ii=1 %to &runs;
* Data management: assign treatment and sequence and period;
data b1;
set b(where=(iteration=%eval(&ii-1)));
subject=_n_;
sequence=mod(_n_,3)+1;
run;

* Transpose the file;
proc sort data=b1;
by sequence subject;
proc transpose data=b1 out=b2(rename=col1=AUC rename=_name_=treat);* drop=_label_);
by sequence subject;
var col1 col2 col3;
run;

* Assign treatment and period variables;
data b2;
    length sequence 3. subject 3. period 3. treat $8.;
    set b2;


    if sequence=1 then
     do;
     * assign period;
        if treat='COL1' then period=1;
           else if treat='COL2' then period=2;
        else period=3;
     end;

    else if sequence=2 then
       do;
       * assign period;
          if treat='COL1' then period=2;
              else if treat='COL2' then period=3;
          else period=1;
     end;
  else
     do;
      * assign period;
        if treat='COL1' then period=3;
           else if treat='COL2' then period=1;
        else period=2;
```

```
     end;

 if treat='COL1' then trt=log(1.0*symget('A'));
 else if treat='COL2' then trt=log(1.0*symget('B'));
 else trt=log(1.0*symget('C'));

 run;

* calculate empirical power;
* use SAS ODS to store the calculated 90% CI;
* Use Proc Mixed procedure to assess dose-prop;

ods listing close;
ODS OUTPUT ESTIMATES=slope;
proc mixed data=b2 method=ml;
        class subject period sequence;
        model auc = trt period sequence/ solution;
        random subject;
        estimate 'slope (beta) and 90% Confidence Interval' trt 1 / alpha=0.10;
        title3 "Dose-proportionality assessment for PK parameter ";
run;

ods exclude none;
ods results;
ods listing;

* Incorporate passed parameters into the final dataset epower;
data epower;
    set slope(keep=estimate Lower Upper rename=(estimate=diff));
    n=symget('N');
    SD=symget('SD');
    runs=symget('runs');
    *alpha=symget('alpha');
run;

* Test the existence of the epower and do the appending;
%if %sysfunc(exist(empower)) %then
    %do;
    * append the data;
    proc append base=empower data=epower force;
    run;
    %end;
%else %do;
    proc datasets library=work nolist;
    change epower = empower;
    run;
    quit;
    %end;

%end;

* Create an indicator variable for calculating empirical power

* calculate the prespecified critical region for beta;
    %let r1=%sysevalf(&rr);
         data temp;
         Theta_L=1.0*symget('Theta_L');
```

```
        Theta_U=1.0*symget('Theta_U');
        ratio=1.0*symget('r1');
        x=1 + log(Theta_L)/log(ratio);
        y=1 + log(Theta_U)/log(ratio);
        call symput('L', put(x,6.4));
        call symput('U', put(y,6.4));
   run;

data prop;
    set empower;
    if _n_=1 then set temp;
    * if indi=1, reject the null hypothesis, otherwise not;
    indi=(lower > x and upper < y );
  run;

* Calculate empirical power;
proc means data=prop noprint;
    var indi;
    output out=propall(drop=_type_ _freq_) mean=power_x;
run;

* Output the Macro parameters;
data power;
    set propall;
    n=symget('N');
    SD=symget('SD');
    meand=1.0*symget('meand');
    runs=symget('runs');
    alpha=symget('alpha');
    lower=symget('lower');
    upper=symget('upper');
    r=symget('r');
run;
%Mend DOSE_PROP3x3;
* The end of this SAS program;
```

**II). Other Crossover Designs**

The SAS macro DOSE_PROP3x3 can be easily adapted to other crossover designs or fixed-sequence designs. For example, for 4x4 design, we only need to change the dimension of the random multivariate normal vector. Some data steps need to be slightly modified.

## RESULTS

Figure 1 shows that the power curve of the 3x3 Latin square design. An 80% power reference line is shown, since dose-proportionality studies generally require the power of at least 80%. The power curve is symmetric and it achieves the maximum when $r^{b_1-1}=1.0$, equivalently, $\beta_1$ takes the value of unit, consistent with the theory of Schuirmann's TOST procedure. It is somewhat risk to use the power given $\beta_1$ is 1.0. In reality, instead of using the power at 1.0 slope value, we usually estimate the power given a small departure from 1.0, such as 1.025, to get a relatively conservative power estimate. Table 1 presents the empirical power estimates for commonly used sample sizes, SDs and criterion boundaries. Table 2 present the sample sizes and corresponding empirical power estimates for at least 80% power. We examined the acceptance region calculated by Eq. (5) with boundaries chosen as ($\theta_L$=0.80, $\theta_H$=1.25).

**Table 1 Empirical Power for 3x3 Latin Square Design with $b_1$ = 1.0 Based on 1000 Simulations**

| | | | Standard Deviation | | |
|---|---|---|---|---|---|
| $(q_L, q_H)$ | Correlation | N | 0.238 | 0.338 | 0.438 |
| (0.80, 1.25) | 0.4 | 16 | 0.93 | 0.57 | 0.21 |
| | | 20 | 0.98 | 0.72 | 0.36 |
| | | 24 | 0.99 | 0.80 | 0.48 |
| | 0.6 | 16 | 0.99 | 0.81 | 0.50 |
| | | 20 | 1.00 | 0.91 | 0.65 |
| | | 24 | 1.00 | 0.95 | 0.73 |
| | 0.8 | 16 | 1.00 | 0.99 | 0.91 |
| | | 20 | 1.00 | 1.00 | 0.95 |
| | | 24 | 1.00 | 1.00 | 0.97 |

**Table 2 Empirical Sample Sizes and Power Based on Power = 80% for 3x3 Latin Square Design with $b_1$ = 1.0 (Number of Monte Carlo Simulations = 1000)**

| | | Standard Deviation | | |
|---|---|---|---|---|
| $(q_L, q_H)$ | Correlation | 0.277 | 0.338 | 0.456 |
| (0.80, 1.25) | 0.4 | 15 (80.7) | 24 (80.9) | 42 (80.2) |
| | 0.6 | 12 (85.4) | 18 (87.0) | 30 (81.6) |
| | 0.8 | 6 (86.6) | 9 (85.4) | 15 (81.8) |
| (0.77, 1.30) | 0.4 | 12 (82.9) | 18 (85.0) | 33 (81.7) |
| | 0.6 | 9 (88.3) | 12 (83.1) | 24 (86.3) |
| | 0.8 | 6 (95.2) | 6 (82.7) | 12 (87.8) |

## CONCLUSION

The Monte Carlo simulation is a useful approach for estimating the power of a variety of statistical tests [3], especially when there are no closed formulae for power calculation. SAS is a practical tool for carrying out this simulation. In this paper, we present a SAS macro to estimate the power of the CI-based dose-proportionality assessment procedure with Latin square crossover designs. The simulation results suggest that the CI-based dose-proportionality assessment procedure provides reasonable power in detecting any departures from proportionality. The sample sizes required to reach a desired power by applying this procedure to a cross-over design are reasonably feasible in practice.

**Figure 1 Empirical Power Curve for the 3x3 Latin Square Design Based on 1000 Simulations**




## REFERENCES


1. Gough K, Hutchison M, Keene O,et al. (1995) Assessment of dose proportionality: report from the pharmaceutical industry. *Drug Information Journal* 29:1039-1048.
2. Smith BP, Vandenhende FR, deSante KA, et al (2000)., Confidence Interval Criteria for Assessment of Dose proportionality. *Pharmaceutical Research*, 17:1278-1283.
3. Zhou J, Coate B, and He Y (2004). Empirical Power Computation Using SAS for Schuirmann's Two One-Sided Tests Procedure in Clinical Pharmacokinetic Drug-Drug Interaction Studies. *SUGI 29 Proceedings*, Paper 154-29.
4. Yuan Y and Zhou J (2005). Cost-efficient higher-order crossover deisgns for two-treatment clinical trials. *Pharmaceutical Statistics*, 4: 245-252.
5. Jones B and Kenward MG (2003). *Design and Analysis of Cross-Over Trials*. Chapman & Hall/CRC.
6. Senn S (2002). *Cross-over Trials in Clinical Research*. John Wiley & Sons, Ltd
7. The MVN Macro, http://ftp.sas.com/techsup/download/stat/mvn.html. The SAS Institute Inc., Cary, NC.


## ACKNOWLEDGMENTS

We are grateful to my clinical and statistical colleagues at Pfizer Inc for valuable suggestions in preparation of this paper. Especially I would like to express great appreciation to Glen Andrews, Randy Allred and Eric Yan for support. This work would not be possible without their support.

## CONTACT INFORMATION

J. Zhou and B. Coate can be reached at Pfizer Inc., 11085 Torreyana Road, San Diego, CA 92121. J. Li can be reached at the Department of Statistics, The University of Michigan, Ann Arbor, MI 48109. M. Chen, Allergan Inc., Irvine, CA 92612.

**ATTACHMENT**

```
*************************************************************************************************************************
*  SAS Program: Power.sas                                                                                               *
*  Function:   Generate empirical power curve plot and report for 3x3 cross-over design                                 *
*              by calling the Macro DOSE_PROP3X3.sas                                                                    *
*                                                                                                                       *
*  Remark:     SAS V8.2                                                                                                 *
*************************************************************************************************************************;
options sasautos=('C:\temp\') nosource nonotes  nosymbolgen nomlogic nomprint nodate
nonumber nocenter;

%Macro POWER(plot=no, table=yes, sampleN=12, alpha=0.05, runs=200, lower=0.80,
upper=1.25, npoints=15, Means=200 400 700, SD=0.356, r=0.40,rr=700/200);

*Delete dataset p_power if any ;
%if %sysfunc(exist(p_power)) %then
      %do;
      Proc datasets library=work nolist;
          delete p_power;
      run;
      quit;
      %end;

%if %upcase(&plot)=YES %then
      %do;
      * The Macro Call to Make Power Curve ***;
      %do number=1 %to &npoints;
      * Create beta for making plots;
         %let plot_mean=log(&lower) + (&number - 1) * (log(&upper)
                        -log(&lower)) /(&npoints - 1);

      * To trick the calculation;
      data _null_;
         x=1+(&plot_mean)/log(&rr);
                call symput('beta',x);
      run;

      %DOSE_PROP3x3(runs=1000,beta=&beta, Means=&Means,
           SD=&SD, N=&sampleN, r=&r,alpha=&alpha, Theta_L=&lower, Theta_U=&upper, rr=&rr)

      data power;
        set power;
        y=&plot_mean;
      run;

      * Test the existence of the power and do the appending;
      %if %sysfunc(exist(p_power)) %then
       %do;
       * append the results;
       proc append base=p_power data=power force;
       run;
       %end;
      %else
              %do;
              * rename power if p_power does not exist;
       proc datasets library=work nolist;
            change power = p_power;
```

```
      run;
      quit;
      %end;
     %end;

     goptions reset=all;
     symbol1 interpol=line value=dot height=1 c=black;
     legend across=1
     cborder=blue
     position=(top inside right)
     mode=share;

     legend2;
     axis1 label=('(beta-1)*log(r)' height=2.5 ) order=(-0.30 to
                0.30 by 0.05)
     minor=(number=1);
     axis2 label=(angle=90 'Empirical power' height=2.5)
     order=(0.0 to 1.0 by 0.1)
     minor=(number=1);
     proc gplot data=p_power;
     plot power_x * y/ haxis=axis1 vaxis=axis2 vref=0.80
                              lvref=2 ;
     run; title; quit;
  %end;

  * delete p_table if any;
  %if %upcase(&table)=YES %then
    %do;
      %if %sysfunc(exist(p_table)) %then
        %do;
          Proc datasets library=work nolist;
            delete p_table;
          run; quit;
        %end;

  * Set n=16 to 24 by 4;
  %do n=16 %to 24 %by 4;
  * Set standard deviation = 0.238 to 0.488 by 0.1;
     %do SD1000=238 %to 438 %by 100;
     * Set correlation coefficient = 0.2, 0.4, 0.6.;
         %do r10=4 %to 8 %by 2;
         * Call Macro DOSE_PROP3x3 to calculate empirical power;
              %DOSE_PROP3x3(runs=&runs,beta=1.0,Means=&Means,SD=%sysevalf(&SD1000/1000),
                    N=&n,r=%sysevalf(&r10/10),alpha=&alpha, Theta_L=&lower,
                    Theta_U=&upper, rr=&rr)

              * Test the existence of the power and do the appending;
              %if %sysfunc(exist(p_table)) %then
                %do;
*append the results;
proc append base=p_table data=power force;
run;
                %end;
              %else
                %do;
  * rename power if p_table does not exist;
```

```
  proc datasets library=work nolist;
      change power = p_table;
  run;
  quit;
%end;
              %end;
            %end;
          %end;

    * make the report;
    data power_all;
      set p_table;
      lcrit=round(lower,0.01);
      ucrit=round(upper,0.01);
      crit="("||trim(left(put(lcrit,4.2)))||",
             "||trim(left(put(ucrit,4.2)))||")";
      _n=input(n,best.);
    run;

    proc sort data=power_all out=power_all;
      by lower _n r sd;
    run;

    proc transpose data=power_all
                   out=xpower(where=(_name_="power_x"));
      by lower _n r;
      var power_x;
      id sd;
      copy crit;
    run;

    ods listing close;
    ods rtf file='C:\temp\table.rtf';
    options nodate nonumber orientation=portrait;
    proc report data=xpower nowindows headline headskip spacing=2 split="@" ;
      columns crit r _n ("SD" _0D238   _0D338  _0D438 );
      define crit /"Criteria" width=16 order;
      define  r /"Correlation" width=16 group;
      define _n /" N " width=10 order=internal ;
      define _0D238 /" 0.238" width=6 f=6.2;
      define _0D338 /" 0.338" width=6 f=6.2;
      define _0D438 /" 0.438" width=6 f=6.2;

      break after crit/skip;
      title; footnote;
    run;
    ods rtf close;
    ods listing;
  %end;

%Mend POWER;
* The end of this Macro;
```